用MYSQL 的DELPHI API 连接操纵MYSQL数据库

MYSQL是一个真正的多用户、多线程、快速、健壮和易用的SQL数据库服务器。如果使用MYSQL 的DELPHI API作为开发接口，可以结合两者的优点：用DELPHI做出漂亮的界面，而MYSQL在后台提供强大的数据存储和处理的引擎。并且没有复杂的ODBC，BDE等中间环节，速度和实用性都得到提高。

首先你要有MYSQL.PAS这个单元文件。可以从http://www.fichtner.net/delphi/mysql/处下载适合你的版本。MYSQL.PAS需要libmySQL.dll的支持。在下面的表中列出了主要的函数功能。

	mysql_init
	获得或初始化一个PMYSQL指针数据。

	mysql_real_connect
	连接一个MySQL服务器。

	mysql_query
	执行指定为一个空结尾的字符串的SQL查询。

	mysql_real_query
	执行指定为带计数的字符串的SQL查询。

	mysql_store_result
	返回由mysql_real_query或mysql_query函数中SQL语句执行的结果。

	mysql_use_result
	初始化一个一行一行地结果集合的检索。  

	mysql_fetch_row
	从结果集合中取得下一行。  

	mysql_fetch_lengths 
	返回当前行中所有列的长度。  

	mysql_num_rows() 
	返回一个结果集合中的行的数量。  

	mysql_affected_rows()  
	返回被最新的SELECT ,UPDATE, DELETE或INSERT查询影响的行数。

	mysql_free_result()
	释放一个结果集合使用的内存。  

	mysql_close()
	关闭指定的连接。


使用MYSQL 的DELPHI API的一般步骤：

1. 使用MYSQL首先要用mysql_init函数初始化并返回一个类型为PMYSQL的指针数据。在以后的mysql_real_connect等函数中都要用到它，是以它为基础进行操作。

2. mysql_real_connect函数连接MYSQL数据库。该函数也返回一个类型为PMYSQL的指针数据，在以后的操作中很多函数（如：mysql_query）都要用到它，它代表了一个活跃的连接。直到mysql_close关闭这个连接。该函数的参数说明如下：

· _mysql是mysql_init函数中初始化好的类型为PMYSQL指针数据。

· 参数host是运行MySQL数据库引擎的机器的TCP/IP主机名，如为NIL则默认为“localhost”；

· 参数user和passwd是MySQL数据库的合法用户和口令； 

· 参数db是连接的数据库名； 

·  参数port，unix_socket和client_flag一般取默认值。 （0，NIL,0）。

3. mysql_query或mysql_real_query将SQL查询发送到MYSQL服务器。它们参数说明如下：

· _mysql是mysql_real_connect函数的返回值。

· q是指向SQL查询的pChar型的字符串。

· length是q所指向的字符串的长度。

4. mysql_store_result函数获得SQL查询的结果，返回一个PMYSQL_RES类型的指针。该函数中用到的参数_mysql是mysql_query或mysql_real_query函数中使用的_mysql。该函数从服务器返回所有SQL查询结果。在调用这个函数后，可以用mysql_field_count()返回字段数量，用mysql_num_rows()返回获得了多少条数据。用mysql_fetch_row () 得到当前行。mysql_fetch_lengths()得到当前行每个字段的长度。

5. mysql_num_fields函数返回一个查询获得的行数。只能在mysql_store_result函数后使用。其参数是mysql_store_result的返回值。

6. mysql_fetch_row函数返回当前行的数据，返回值为一个PMYSQL_ROW类型数据。如果没有数据返回,或已经到达数据库的尾部，其返回值为NIL。其参数是mysql_store_result或mysql_use_result返回PMYSQL_RES类型的指针。这里重点介绍一下PMYSQL_ROW数据结构，其在MYSQL.PAS是这样定义的：

PMYSQL_ROW = ^TMYSQL_ROW;  // return data as array of strings


TMYSQL_ROW = array[0..MaxInt div SizeOf(pChar) - 1] of pChar;

可以知道其实它指向一个指针数组，数组的实际大小由mysql_num_fields()可知。在MYSQL中除了CHAR类型的数据用pChar(指向以0为结尾的指针)返回外，象DOUBLE、INT等类型数据也用字符串的方式返回。而对于返回值是二进制数据象BLOB等也用pChar指向返回的数据地址，在Delphi中要用PbyteArray等指针类型对他们进行类型强制转换，然后再使用。每个指针指向的数据区大小由mysql_fetch_lengths()返回。

7. 查询结果用完后，使用mysql_free_result()函数释放内存。其参数是mysql_store_result()函数返回的PMYSQL_RES类型的指针。

8. 和MYSQL会话结束后，要用mysql_close()关闭连接。其参数是mysql_real_connect()函数的返回值。

下面通过一个实例来说明API接口的用法。

有一个学生数据库(STUDENT)，其中有一个表(student_list)，表的结构如下：

+---------+--------------------+------+-----+---------+-------+

| Field   | Type               | Null | Key | Default | Extra |

+---------+--------------------+------+-----+---------+-------+

| name    | varchar(20) binary | YES  |     | NULL    |       |

| sex     | char(2) binary     | YES  |     | NULL    |       |

| birth   | date               | YES  |     | NULL    |       |

| picture | blob               | YES  |     | NULL    |       |

+---------+--------------------+------+-----+---------+-------+

其中picture中存储了学生的照片，是jpg格式的。程序中把此表中的前三个字段的内容独到一个表格中，并显示其照片。其窗体如下图：

源程序如下：

unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, ComCtrls, Grids, StdCtrls, ExtCtrls,mysql ,jpeg;

type

  TForm1 = class(TForm)

    Image1: TImage;

    s_list: TStringGrid;

    Edit1: TEdit;

    Label1: TLabel;

    Edit2: TEdit;

    Label2: TLabel;

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormCreate(Sender: TObject);

    procedure FormClose(Sender: TObject; var Action: TCloseAction);

    procedure s_listRowMoved(Sender: TObject; FromIndex, ToIndex: Integer);

    procedure s_listSelectCell(Sender: TObject; ACol, ARow: Integer;

      var CanSelect: Boolean);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

  start_mysql:PMYSQL;

  rowi: PMYSQL_ROW;

  this_res:PMYSQL_RES;

  lengths:PMYSQL_LENGTHS;

  old_row:integer;

procedure image_show();//

  implementation

{$R *.dfm}

procedure image_show();

var JPEGImagei : TJPEGImage;

    graph_stream:TMemoryStream;

    p_jpg:PByteArray
;

begin

try

  JPEGImagei:= TJPEGImage.Create;

  graph_stream:=TMemoryStream.Create;

  graph_stream.SetSize(lengths^[3]);//取得blob字段的长度

  p_jpg:=PByteArray(rowi^[3]);

  graph_stream.writeBuffer(p_jpg^,lengths^[3]);

  graph_stream.Seek(0,0);

  JPEGImagei.LoadFromStream(graph_stream);

  form1.image1.Picture.Assign(JPEGImagei);//显示照片

finally

  JPEGImagei.Free;

graph_stream.Free;

end;

end;

procedure TForm1.Button1Click(Sender: TObject);

var i,j,row_num:integer;

begin

  start_mysql:=mysql_init(start_mysql);

  if start_mysql=NIL then

      MessageDlg('系统初始化失败',mtWarning,[mbOK],0)

  else

  begin

start_mysql:=mysql_real_connect(start_mysql,pchar('localhost'),pchar(edit1.Text),pchar(edit2.Text),pchar('student'),0,nil,0);

if start_mysql=nil then  MessageDlg('连接数据库失败，请检查mysql\data\下的student子目录。',mtWarning,[mbOK],0)

    else

    begin

      if mysql_query(start_mysql,PCHAR('select * from student_list'))=0 then

      begin

        this_res:=mysql_store_result(start_mysql);

        row_num:=mysql_num_rows(this_res);

        s_list.RowCount:=row_num+1;

        for i:=1 to row_num do

        begin

          rowi:=mysql_fetch_row(this_res);

          lengths:= mysql_fetch_lengths(this_res);

          s_list.Cells[0,i]:=inttostr(i);

          for j:=1 to 3 do

            s_list.Cells[j,i]:=string(rowi^[j-1]);//在表格中显示信息

          if i=1 then image_show();

          s_list.Row:=1;

          old_row:=1;

        end;

      end

      else

         MessageDlg('SQL查询失败，请确认数据表STUDENT_LIST存在',mtWarning,[mbOK],0)

    end;

  end;

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

  s_list.Cells[0,0]:='序号';

  s_list.Cells[1,0]:='姓名';

  s_list.Cells[2,0]:='性别';

  s_list.Cells[3,0]:='出生日期';

end;

procedure TForm1.FormClose(Sender: TObject; var Action: TCloseAction);

begin

  if this_res<>nil then mysql_free_result(this_res);

  if start_mysql<>nil then  mysql_close(start_mysql);

end;

procedure TForm1.s_listRowMoved(Sender: TObject; FromIndex,

  ToIndex: Integer);

begin

  if this_res<>nil then

  begin

  end;

end;

procedure TForm1.s_listSelectCell(Sender: TObject; ACol, ARow: Integer;

  var CanSelect: Boolean);

begin

  if (this_res<>nil)and(old_row<>AROW) then

  begin

    mysql_data_seek(this_res,AROW-1);

    rowi:=mysql_fetch_row(this_res);

    lengths:= mysql_fetch_lengths(this_res);

    image_show();

    old_row:=AROW;

  end;

end;

end.
程序运行后，单击“打开学生档案”按钮，即可浏览学生档案数据库，此程序在mysql 3.23.36 + Delphi 5.0在本机上调试通过。

